
Introduction

The CKPMpp pre-processor transforms source files into compilation units similar to [1].
1. EBNF of the Pre-Processor
File
=
{ Chunk }.

Chunk
=
NewLine { WhiteSpace } PreProcLine  { WhiteSpace }




{ Comment {WhiteSpace } } NewLine.



| Comment



| AllCharacter.
Comment
=
"/*" { AllCharacter } "*/".
PreProcLine
=
"#" { WhiteSpace } ( include-Statement | if-Statement | ifdef-Statement | else-Statement


| elif-Statement | endif-Statement | define-Statement | defined-Statement)
include-Statement
=
"include" Separator ( "\"" path "\"" | "<" path ">" ).

if-Statement
=
"if" Separator LogicalOrExpression.

ifdef-Statement
=
"ifdef" Separator LogicalOrExpression.

defined-Statement
=
"defined" Separator MacroIdentifier

else-Statement
=
"else".

elif-Statement
=
"elif" Separator LogicalOrExpression.

endif-Statement
=
"endif".

define-Statement
=
"define" Separator MacroIdentifier Separator SimpleExpression

LogicalOrExpression
=
LogicalAndExpression { "||" LogicalAndExpression }.

LogicalAndExpression
=
CompExpression { "&&" CompExpression }.

CompExpression
=
SimpleExpression [ ( "==" | "!=" | ">=" | ">" | "<" | "=<" ) SimpleExpression ]



| defined-Statement.

SimpleExpression
=
[ UnaryOperator ] Term { ( "+" | "-" | "|" | "<<" | ">>" ) Term }.

UnaryOperator
=
( "+" | "-" )

Term
=
Factor { ( "*" | "/" | "%" | "!" ) Factor }.

Factor
=
"(" LogicalOrExpression ")" | MacroIdentifier | Constant.

Constant
=
Number | QuotedString | String

QuotedString
=
"\"" ( "\\" AllCharacter | NotQuoteCharacter ) "\""

Separator
=
WhiteSpace { WhiteSpace }

String
=
Character { Character }.

NotQuoteCharacter
=
' ' .. '!' | '#' .. '~'

AllCharacter
=
NewLine | WhiteSpace | Character

Character
=
Digit | Letter | SpecialCharacter.

Number
=
Digit { Digit }.

Digit
=
'0' .. '9'.

NewLine
=
'\r' | '\n' | '\f'.

WhiteSpace
=
'\t' | ' '.

Letter
=
'A' .. 'Z' | 'a' .. 'z'.

EscapeCharacter
=
'\\'.

SingleQuote
=
'\''.

DoubleQuote
=
'"'.

SpecialCharacter
=
'!' | '#' | '$' | '%' | '&' | '(' | ')' | '*' | '+' | ',' | '-' | '.' | '/' | ':' | ';' | '<' | '=' | '>' | '?'



| '[' | '\' | ']' | '^' | '_' | '`' | '{' | '|' | '}' | '~'

